SR 122

Special Report 122

DIGITAL SOLUTION OF MODIFIED BERGGREN EQUATION TO CALCULATE DEPTHS OF FREEZE OR THAW IN MULTILAYERED SYSTEMS

George W. Aitken
and
Richard L. Berg

October 1968

CONDUCTED FOR
CORPS OF ENGINEERS, U.S. ARMY
BY
U.S. ARMY MATERIEL COMMAND
TERRESTRIAL SCIENCES CENTER
COLD REGIONS RESEARCH & ENGINEERING LABORATORY
HANOVER, NEW HAMPSHIRE

Special Report 122

DIGITAL SOLUTION OF MODIFIED BERGGREN EQUATION TO CALCULATE DEPTHS OF FREEZE OR THAW IN MULTILAYERED SYSTEMS

George W. Aitken
and
Richard L. Berg

October 1968

CONDUCTED FOR
CORPS OF ENGINEERS, U.S. ARMY
BY
U.S. ARMY MATERIEL COMMAND
TERRESTRIAL SCIENCES CENTER
COLD REGIONS RESEARCH & ENGINEERING LABORATORY
HANOVER, NEW HAMPSHIRE

PREFACE

Authority for the investigation reported herein is contained in FY 1967 Instructions and Outline, Military Construction Investigations, Engineering Criteria and Investigations and Studies, Investigation of Arctic Construction: Thermal Calculation Techniques.

The Military Construction Investigations program is conducted for the Engineering Division, Directorate of Military Construction, Office, Chief of Engineers and is administered by the Civil Engineering Branch (Mr. T.B. Pringle, Chief). This study was conducted by the Cold Regions Research and Engineering Laboratory (CRREL) of the U.S. Army Terrestrial Sciences Center.

Investigations were performed under the general supervision of Mr. K.A. Linell, Chief, Experimental Engineering Division, CRREL, and the direct supervision of Mr. E.F. Lobacz, Chief, Construction Engineering Branch, CRREL. The report was prepared by Messrs. G.W. Aitken and R.L. Berg, Research Civil Engineers, Construction Engineering Branch.

Lieutenant Colonel John E. Wagner was Commanding Officer/Director of the U.S. Army Terrestrial Sciences Center during the publication of this report, and Mr. W.K. Boyd was Chief Engineer.

The U.S. Army Terrestrial Sciences Center is a research activity of the Army Materiel Command.

CONTENTS

ILLUSTRATIONS

TABLES

ABSTRACT

This report presents a method for a digital computer solution, using the FORTRAN language, of the modified Berggren equation for computing depths of frost and thaw penetration in non-homogeneous (multilayered) soil systems. A program source listing, sample solutions, and tables of thermal properties of soils and construction materials are presented.

NOMENCLATURE

Item	Symbol	Program	Units	Multiply by	To obtain
Index (freeze or thaw)	I	DEGREE	degree-days F	$\frac{5}{9}$	degree-days C
n-factor	n	EN			
Length of season	t	T	days		
Average surface temp differential	v_s	VS	°F	$\frac{5}{9}$	°C
Initial temp difference	v_0	VO	°F	$\frac{5}{9}$	°C
Thermal ratio	α	ALPHA			
Mean annual temp	M.A.T.	ATM	°F	$\frac{5}{9}$(°F-32)	°C
Dry unit weight	γ_d	GAMMA	lb/cu ft	16.0185	kg/cu m
Moisture content	w	WC	% dry wt		
Thickness	d	D	ft	0.3048	m
Volumetric heat capacity	C	C	Btu/cu ft °F	16.018	kg cal/cu m °C
Thermal conductivity	K	THERM	Btu/ft hr °F	1.488	kg cal/m hr °C
Volumetric latent heat of fusion	L	FUSION	Btu/cu ft	8.899	kg cal/cu m
Thermal diffusivity	a	A	sq ft/hr	0.2581	sq cm/sec
Fusion parameter	μ	U			
Lambda coefficient	λ	F			
Thermal resistance		RES	sq ft hr °F/Btu	2045	sq cm hr °C/kg cal
Material type		KIND			
Thickness summation		SUMD	ft	0.3048	m
Total latent heat		SUMFD	Btu/sq ft	2.71	kg cal/sq m
Average latent heat		BARL	Btu/cu ft	8.899	kg cal/cu m
Total volumetric heat		SUMCD	Btu/sq ft °F	4.892	kg cal/sq m °C
Average volumetric heat		BARC	Btu/cu ft °F	16.018	kg cal/cu m °C
Thermal resistance above layer		SUMRES	sq ft hr °F/Btu	2045	sq cm hr °C/kg cal
Total thermal resistance		TOTALR	sq ft hr °F/Btu	2045	sq cm hr °C/kg cal
Index used in layer		AYERI	degree-days F	$\frac{5}{9}$	degree-days C
Summation of index		SI	degree-days F	$\frac{5}{9}$	degree-days C
Layer counter		N			
Program branch control		NNN			
$SUMD_{(n-1)}$		CON	ft	0.3048	m
$SI_{(n-1)}$		SIN	degree-days F	$\frac{5}{9}$	degree-days C
Error function		ERF			
Complementary error function		ERFC			
Exponential function		EXP			
Computed variable (Defined by Aldrich and Paynter (1953), eq A-13, p. A-4.)	Z	R			

DIGITAL SOLUTION OF MODIFIED BERGGREN EQUATION TO CALCULATE DEPTHS OF FREEZE OR THAW IN MULTILAYERED SYSTEMS

by

George W. Aitken and Richard L. Berg

Introduction

This report details the procedures used to prepare a computer program for solution of the modified Berggren equation for calculating the depth of freeze or thaw in a multilayered soil system and presents the information the reader needs to use this program.

Some of the more important data used to prepare the modified Berggren solution, together with thermal conductivity curves after Kersten (1949), are included to minimize the need for additional reference material. A description of the more important computation techniques and a discussion of the limitations of this particular solution are also included.

Appendix A contains a complete FORTRAN II program with operating instructions and Appendix B contains input and output data for two typical solutions.

General

The method of solution reported here is presently used by the Departments of the Army and the Air Force (1966) and others (see Sanger, 1963) to determine one of the factors relating to the design depths of granular materials beneath roadways and airfields. Solution of the modified Berggren equation for a multilayered system is very laborious by hand while the general case is readily adapted to computer solution.

The modified Berggren equation (eq 1) was developed by Aldrich and Paynter (1953) under a contract with the former Arctic Construction and Frost Effects Laboratory (ACFEL).*

$$x = \lambda \sqrt{\frac{48\,K\,n\,I}{L}} \qquad (1)$$

where x = depth of frost or thaw penetration

λ = lambda coefficient

K = average thermal conductivity

n = n-factor to convert an air index to a surface index

I = air freezing or thawing index

L = volumetric latent heat of fusion.

A complete development of this equation and a discussion of the necessary assumptions and simplifications made during its development are contained in Aldrich and Paynter (1953). A few of the more important assumptions and some of the limitations of the equation and of this particular solution are discussed below. The assumptions and limitations apply regardless of whether the equation is used to determine the depth of freeze or the depth of thaw, so only the freezing model is discussed.

*ACFEL was merged with the former Snow, Ice and Permafrost Research Establishment (SIPRE) in 1961 to form the U.S. Army Cold Regions Research and Engineering Laboratory.

Assumptions. The mathematical model assumes one-dimensional heat flow with the entire soil mass at its mean annual temperature prior to the start of the freezing season. It is assumed that when the freezing season starts, the surface temperature changes suddenly (as a step function) from the mean annual temperature to a temperature v_s degrees below freezing and remains at this new temperature throughout the entire freezing season. The effect of latent heat is considered as a heat sink at the moving frost line, and it is assumed that the soil freezes at a temperature of 32.0F.

Limitations. The model on which the modified Berggren equation is based further assumes an isothermal system at the beginning of the season and therefore cannot normally be used to calculate thaw depths in seasonal frost areas or frost depths in permafrost areas. The equation cannot successfully be used to calculate penetration over part of the season. Aldrich and Paynter (1953) state, "Attempts to calculate the depth-time curves based on partial freezing indices are likely to encounter substantial errors." Experience has shown this to be true.

The Lambda coefficient (Aldrich and Paynter, 1953) used in this solution is obtained from eq 2:

$$\lambda = \sqrt{2Z^2/\mu} \tag{2}$$

where Z is obtained from the following transcendental equation. The subscripts t and f indicate thawed and frozen values respectively.

$$\frac{\exp(-Z^2)}{\operatorname{erf}(Z)} - \left(\frac{a_f}{a_t}\right)^{1/2}\left(\frac{K_t}{K_f}\right)^{1/2}\frac{v_0}{v_s}\,\frac{\exp\left(-\frac{a_f}{a_t}Z^2\right)}{\operatorname{erfc}\left(\frac{a_f}{a_t}\right)^{1/2}Z} = \frac{Z L \pi^{1/2}}{C v_s} \tag{3}$$

and $\mu = v_s C_f / L$.

Aldrich and Paynter averaged the thawed and frozen soil thermal properties in their solutions and the same assumption is made herein. Therefore we assume that

$$a_f = a_t$$

$$K_f = K_t$$

and using the relationships

$$a = v_0/v_s$$

$$\operatorname{erfc}(x) = 1 - \operatorname{erf}(x)$$

equation 3 may be reduced to

$$\frac{\exp(-Z^2)}{\operatorname{erf}(Z)} - \frac{a\exp(-Z^2)}{1-\operatorname{erf}(Z)} = \frac{Z L \pi^{1/2}}{C v_s}. \tag{4}$$

This transcendental equation is solved by iteration with the error function, computed accurate to the sixth place after the decimal, obtained from the following equation (Hastings, 1955):

$$\mathrm{erf} = 1 - 1/(1 + a_1x + a_2x^2 + a_3x^3 + a_4x^4 + a_5x^5 + a_6x^6)^{16} \quad (5)$$

where $a_1 = .0705230784$ $a_4 = .0001520143$

$a_2 = .0422820123$ $a_5 = .0002765672$

$a_3 = .0092705272$ $a_6 = .0000430638.$

Program details

The program uses 5000 words of memory and was written for a standard Honeywell DDP-24 computer using the input-output typewriter, paper tape reader and high speed printer. Major data input to the program is by perforated tape with provision for typewriter entry of volumetric heat capacity (C), thermal conductivity (THERM) and latent heat (FUSION) for any layer.

Input data preparation

For all solutions certain basic data are required by the program. These data, on perforated paper tape, are the first information entered into the computer for each solution:

Freezing or thawing index (DEGREE)

n-factor (EN)

Length of season (T)

Mean annual temperature (ATM).

All data are in real notation terminated with an end-of-word character.

The n-factors for air freezing conditions are given in Table I and the relationship between wind speed and the n-factor for pavements during the thawing season is contained in Figure 1. n-factors of 1.0 and 2.0 are suggested for turf and gravel surfaces, respectively, under thawing conditions. If the surface index is used as input data, an n-factor of 1.0 should be used.

Table I. n-factor for freezing conditions.

(Dept. of the Army and the Air Force, 1966)

Surface type	*n-factor*
Snow	1.0
Pavements free of snow and ice	0.9
Sand and gravel	0.9
Turf	0.5



Figure 1. Relationship between wind speed and n-factor during thawing season (after Sanger, 1963).

The data for each layer in the problem may be entered in one of two forms:

1. All data on perforated tape in the sequence shown, and in either of the following formats:

   A. Material type (KIND) = 1. or 2.
      Density (GAMMA)
      Water content (WC)
      Layer thickness (D)

   B. Material type (KIND) = 10.
      Density (GAMMA)
      Water content (WC)
      Layer thickness (D)
      Volumetric heat capacity (C)
      Thermal conductivity (THERM)
      Latent heat (FUSION)

2. Certain thermal parameters for type 10. materials (see Table III for material types) may be entered on the typewriter in which case the tape format for the layer would be as follows:

   Material type (KIND) = 10.
   Density (GAMMA)
   Water content (WC)
   Layer thickness (D)

Table II and Figures 2-9 will aid in choosing thermal properties for type 10. materials.



Figure 2. Average thermal conductivity, sandy soils, frozen (after Kersten, 1949).



Figure 3. Average thermal conductivity, sandy soils, unfrozen (after Kersten, 1949).



Figure 4. Average thermal conductivity, silt and clay soils, frozen (after Kersten, 1949).



Figure 5. Average thermal conductivity, silt and clay soils, unfrozen (after Kersten, 1949).



Figure 6. Average thermal conductivity, peat, frozen (after Kersten, 1949).



Figure 7. Average thermal conductivity, peat, unfrozen (after Kersten, 1949).



Figure 8. Average volumetric heat capacity for soils (after Aldrich and Paynter, 1953).



Figure 9. Volumetric latent heat for soils (after Aldrich and Paynter, 1953).

Table II. Thermal properties of construction materials (from Dept. of the Army and the Air Force, 1966).

Type of material	Description	Unit weight lb./cu. ft.	k Conductivity B.t.u./sq. ft. hr. °F./in.	K Conductivity B.t.u./ft. hr. °F.
Asphalt paving mixture	Mix with 6% by weight cut-back asphalt	138	10.3	0.86
Concrete[1]	With sand and gravel or stone aggregate (oven-dried)	140	9.0	0.75
	With sand and gravel or stone aggregate (not dried)	140	12.0	1.00
	With lightweight aggregates including expanded shale, clay, or slate; expanded slags; cinders; pumice; perlite; vermiculite; also cellular concretes.	120	5.2	0.43
		100	3.6	0.30
		80	2.5	0.21
		60	1.7	0.14
		40	1.15	0.096
		30	0.90	0.075
		20	0.70	0.058
Wood[1]	Maple, oak, and similar hardwoods	45	1.10	0.092
	Fir, pine, and similar softwoods	32	0.80	0.067
Building boards[1]	Asbestos-cement board	120	4.0	0.33
	Plywood	34	0.80	0.067
	Wood fiberboard, laminated or homogeneous	26, 33	0.42, 0.55	0.035, 0.046
	Wood fiber--hardboard type	65	1.40	0.12
Insulating materials				
Blanket and batt[1]	Mineral wool, fibrous form, processed from rock, slag, or glass	1.5-4.0	0.27	0.022
	Wood fiber	3.2-3.6	0.25	0.021
Board and slabs[1]	Cellular glass	9.0	[2]0.39	0.032
	Corkboard (without added binder)	6.5-8.0	[2]0.27	0.022
	Glass fiber	9.5-11.0	0.25	0.021
	Wood or cane fiber--interior finish (plank, tile, lath)	15.0	0.35	0.029
	Expanded polystyrene	1.6	0.29	0.024
	Mineral wool with resin binder	15	[2]0.28	0.023
	Mineral wool with asphalt binder	15	[2]0.31	0.026
Loose fill	Mineral wool (glass, slag, or rock)	2.0-5.0	0.30	0.025
	Sawdust or shavings	8.0-15.0	0.45	0.037
	Vermiculite (expanded)	7.0-8.2	0.48	0.040
	Wood fiber: redwood, hemlock, or fir	2.0-3.5	0.30	0.025
Miscellaneous	Water, average	62.4	4.2	0.35
	Ice	57	[3]15.4	1.28

[1] Values extracted from ASHRAE Guide and Data Book, 1963, by permission. Values for k are for dry building materials at a mean temperature of 75° F. except as noted; wet conditions will adversely affect values of many of these materials.

[2] Mean temperature of 60° F.

[3] Mean temperature of 32° F.

Table III. Material types.

Kind	*Description*
1.	Granular soils (gravels and sands)
2.	Fine-grained soils (silts and clays)
10.	Other (pavements, insulations, organic soils, etc.)

The average volumetric heat capacity and latent heat for materials other than granular or fine-grained soils may be obtained from Figures 8 and 9, respectively, or from eq 6 and 7:

Average volumetric heat capacity

$$C = \gamma_d[0.17 + 0.75(w)]. \tag{6}$$

Latent heat

$$L = 144(\gamma_d)(w). \tag{7}$$

Appendix B contains the input data and the typewriter and printer output for two typical solutions.

Note that the same basic input data give slightly different results in the two sample solutions because of a difference in computational method. This difference is insignificant, about 1.5%, and is caused by the difference in the thermal conductivity (THERM) values calculated in the program (SOLN. 2) versus those entered manually (SOLN. 1).

Comments on computer solution

This program cannot be used for solution of the modified Berggren equation for a homogeneous case. Equation 1 should be used for this type of solution.

The index printed by the program is the surface index. To obtain the air index, the surface index must be divided by the n-factor.

Literature cited

Aldrich, H.P. and Paynter, H.M. (1953) Analytical studies of freezing and thawing of soils, First interim report. U.S. Army Corps of Engineers, New England Division, Arctic Construction and Frost Effects Laboratory (ACFEL) Technical Report 42.

Hastings, C., Jr. (1955) *Approximation for digital computers.* Princeton, N.J.: Princeton University Press.

Kersten, M.S. (1949) Laboratory research for the determination of the thermal properties of soils, Final report. U.S. Army Corps of Engineers, New England Division, ACFEL Technical Report 23.

Departments of the Army and Air Force (1966) Arctic and subarctic construction, Calculation methods for determination of depths of freeze and thaw in soils. Department of the Army Technical Manual TM-5-852-6. Department of the Air Force Manual AFM 88-19, Chap. 6.

Sanger, F.J. (1963) Degree-days and heat conduction in soils. Proceedings of the Permafrost International Conference, National Academy of Science-National Research Council Publication No. 1287.

```
C       MODIFIED BERGGREN SOLUTION FOR MULTILAYERED SYSTEMS
C       G. AITKEN/R. BERG, MARCH,1966
C       LATEST UPDATE 5/9/68, GWA
C       EDITING REVISIONS FOR SR-122
C
C                            OPERATING INSTRUCTIONS
C
C       A.  PLACE DATA TAPE PREPARED IN ACCORDANCE WITH INSTRUCTION
C       MANUAL IN PHOTOREADER.
C       B.  SET SENSE SWITCH 1 ON FOR FROST DEPTH SOLUTION, OFF
C       FOR THAW DEPTH SOLUTION.
C       C.  SET SENSE SWITCH 2 ON FOR TAPE INPUT OF C, THERM AND
C       FUSION, OFF FOR KEYBOARD INPUT.
        REAL KIND
        A1=.0705230784
        A2=.0422820123
        A3=.0092705272
        A4=.0001520143
        A5=.0002765672
        A6=.0000430638
        TYPE 510
510     FORMAT(1H ,53HSENSE SWITCH SETTINGS FOR THIS PROGRAM ARE AS
       1 FOLLOWS,/,
       21H ,49H1 ON FOR FROST DEPTH SOLUTION, OFF FOR THAW DEPTH,/,
       11H ,66H2 ON FOR TAPE INPUT OF C, THERM AND FUSION, OFF FOR
       2 KEYBOARD INPUT,/,
       31H ,74HPLACE DATA TAPE IN PHOTOREADER AND PRESS START
       4 TO INITIATE FIRST SOLUTION.)
        PAUSE 1
101     PRINT 200
200     FORMAT(1H1,25X,53HMULTILAYER SOLUTION OF THE MODIFIED
       1BERGGREN EQUATION///)
        PRINT 201
201     FORMAT(1H ,11X,3HDRY,4X,5HWATER,4X,5HLAYER,4X,4HHEAT,4X,7HTHERMAL,
       12X,6HLATENT,7H FUSION,7H LAMBDA,2X,5HLAYER,3X,4HCUM.,5X,5HLAYER,4X
       2,4HCUM.,/,10X,7HDENSITY,8H CONTENT,3X,4HSIZE,3X,8HCAPACITY,3X,5HCO
       3ND.,4X,4HHEAT,3X,4HPAR.,11X,4HRES.,3X,4HRES.,5X,5HINDEX,3X,5HINDEX
       4,/,10X,7H(GAMMA),2X,4H(WC),6X,3H(D),5X,3H(C),5X,7H(THERM),9H (FUSI
       5ON),2X,3H(U),3X,3H(F),4X,5H(RES),9H (SUMRES),2X,7H(AYERI),3X,4H(SI
       6),/)
        DEPTHO=0.
        DEPTH=0.
        SUMD=0.
        SUMFD=0.
        SUMCD=0.
        SUMIND=0.
        RES=0.
        SUMRES=0.
        NNN=10
        N=0
        SI=0.
        READ PAPER TAPE 1, DEGREE,EN,T,ATM
1       FORMAT(4E20.8)
        DEGREE=ABSF(DEGREE)
        VO=ABSF(ATM-32.)
        VS=EN*DEGREE/T
        DEGREE=DEGREE*EN
        ALPHA=VO/VS
2       FLAG=-100.
        FLAGA=-100.
        READ PAPER TAPE 1, KIND,GAMMA,WC,D
        IF(KIND-10.)4,5,5
4       IF(KIND-02.)6,7,7
6       THERMT=(.7*FLOG10F(WC)+.4)*(10.**(.01*GAMMA))
        THERMF=.076*(10.**(.013*GAMMA))+.032*(10.**(.0146*GAMMA))*WC
        THERM=(THERMT+THERMF)/24.
        GO TO 20
7       THERMT=(.9*FLOG10F(WC)-.2)*(10.**(.01*GAMMA))
        THERMF=.01*(10.**(.022*GAMMA))+.085*(10.**(.008*GAMMA))*WC
        THERM=(THERMF+THERMT)/24.
20      C=GAMMA*(.17+(.75*WC/100.))
        FUSION=144.*GAMMA*(WC/100.)
        GO TO 21
5       IF(SENSE SWITCH 2)220,217
217     TYPE 208
208     FORMAT(1H ,36HTYPE C, TAB, THERM, TAB, FUSION, CR.)
209     ACCEPT 9, C,THERM,FUSION
9       FORMAT(3E20.8)
        GO TO 21
220     READ PAPER TAPE 9, C,THERM,FUSION
21      A=THERM/C
        SUMD=SUMD+D
        DEPTH=SUMD
10      SUMFD=SUMFD+(FUSION*D)
        BARL=SUMFD/SUMD
        SUMCD=SUMCD+(C*D)
        BARC=SUMCD/SUMD
        IF(BARL)37,37,36
36      U=VS*(BARC/BARL)
```

```
      GARB=1.77245/U
      R=.15
30    ERF=1.-1./(((1.+A1*R+A2*R**2+A3*R**3+A4*R**4+A5*R**5
     1  +A6*R**6)**16)
      EXP=EXPF(-(R**2))
      ERFC=1.-ERF
      CHECK=EXP/ERF-((ALPHA)*EXP/ERFC)
      CHECKA=R*GARB
      IF(FLAG) 1002,1002,1003
1002  FLAG=+100.
      IF (CHECK-CHECKA)1004,1000,1000
1004  FLAGA=+100.
      GO TO 1001
1003  IF(FLAGA)1000,1000,1001
1000  IF(CHECK-CHECKA)35,35,34
1001  IF(CHECK-CHECKA)1007,35,35
1007  R=R-.001
      GO TO 30
34    R=R+.001
      GO TO 30
35    F=((2.*(R**2))/U)**.5
      GO TO 38
37    F=0.
      U=0.
38    RES=ABSF(D)/THERM
      TOTALR=SUMRES+RES/2.
      AYERI=(FUSION*(ABSF(D))/(24.*(F**2)))*TOTALR
      SI=SI+AYERI
      N=N+1
      PRINT 11,N,GAMMA,WC,D,C,THERM,FUSION,U,F,RES,SUMRES,AYERI,SI
11    FORMAT(3X,I3,2X,2F8.2,2X,F7.3,2X,F6.2,2X,F8.4,2X,F6.0,2X,F5.2,
     1  2X,F4.2,2X,F6.2,2X,F6.2,2X,F7.0,2X,F7.0)
      IF(NNN)15,12,12
12    IF(DEGREE-SI)13,100,14
14    IF((DEGREE-SI)-10.)100,100,501
501   SUMRES=SUMRES+RES
      GO TO 2
13    CON=SUMD-D
      DEPTHO=DEPTH-D
      SIN=SI-AYERI
      NNN=-10
15    IF(DEGREE-SI)17,100,16
16    IF(DEGREE-(SI+10.))100,100,18
18    SUMFD=SUMFD-(FUSION*D)
      SUMCD=SUMCD-(C*D)
      D=D*((DEGREE-SIN)/AYERI)
      SUMD=CON+D
      SI=SIN
      GO TO 10
17    IF((DEGREE+10.)-SI)18,100,100
100   PRINT 701, DEGREE
701   FORMAT(1H ,////,5X,16HSURFACE INDEX = ,F7.0,1X,13HDEGREE DAYS F)
      PRINT 702, EN
702   FORMAT(1H ,4X,11HN FACTOR = ,F5.2)
      PRINT 703, T
703   FORMAT(1H ,4X,19HLENGTH OF SEASON = ,F5.0,1X,4HDAYS)
      PRINT 704, ATM
704   FORMAT(1H ,4X,19HMEAN ANNUAL TEMP = ,F5.1,0X,1HF/)
      IF(SENSE SWITCH 1)32,33
32    PRINT 202, SUMD
202   FORMAT(1H ,4X,47HDEPTH OF FROST PENETRATION FOR THIS SOLUTION
     1 = F4.1,1X,5HFEET.)
      GO TO 500
33    PRINT 203, SUMD
203   FORMAT(1H ,4X,46HDEPTH OF THAW PENETRATION FOR THIS SOLUTION
     1 = F4.1,1X,5HFEET.)
500   TYPE 204
204   FORMAT(1H ,95HSOLUTION COMPLETE. PLACE NEW DATA TAPE IN
     1 PHOTOREADER AND PRESS START TO INITIATE NEW SOLUTION.)
      PAUSE 7
      GO TO 101
      END
```

SIMPLIFIED FLOW DIAGRAM, BERGGREN MULTILAYER SOLUTION




C
COMPUTE A, SUMD, DEPTH
E
COMPUTE SUMFD, BARL, SUMCD, BARC
BARL
>
≤
COMPUTE F, U
F = 0
U = 0
COMPUTE RES, TOTALR, AYERI, SI
N = N + 1
PRINT N, GAMMA, WC, D, THERM, FUSION, U, F, RES, SUMRES, AYERI, SI
NNN
<
≥
DEGREE-SI
<
=
>
COMPUTE DEPTHO, CON, SIN
NNN = −10
(DEGREE-SI):10
≤
>
PAUSE SOLUTION COMPLETE
D
A



NOTES:

1. Method of computing F is described in text.
2. Sense Switch 1 is used only for output format control and is not shown on diagram.

APPENDIX B. SAMPLE SOLUTIONS

SOLUTION 1

Tape input of all data

INDEX = 780.
EN = 2.
T = 105.
ATM = 12.

LAYER	1	2	3	4	5
KIND	11.	12.	13.	14.	15.
GAMMA	138.	156.	151.	130.	122.
WC	0.	2.1	2.8	6.5	4.6
D	.4	1.6	3.0	1.0	2.0
C	28.	29.	29.	28.	25.
THERM	.86	1.85	2.0	1.65	.64
FUSION	0.	470.	610.	1220.	808.

SOLUTION 2

Tape and typewriter data input

LAYER	1	2	3	4	5
KIND	11.	1.	1.	1.	2.
GAMMA	138.	156.	151.	130.	122.
WC	0.	2.1	2.8	6.5	4.6
D	.4	1.6	3.0	1.0	2.0
C	28.^{T}				
THERM	.86^{T}				
FUSION	0.^{T}				

T = Typewriter input

Sample Solution No. 1

Typewriter output:

```
SENSE SWITCH SETTINGS FOR THIS PROGRAM ARE AS FOLLOWS
1 ON FOR FROST DEPTH SOLUTION, OFF FOR THAW DEPTH
2 ON FOR TAPE INPUT OF C, THERM AND FUSION, OFF FOR KEYBOARD INPUT
PLACE DATA TAPE IN PHOTOREADER AND PRESS START TO INITIATE FIRST SOLUTION.


SOLUTION COMPLETE, PLACE NEW DATA TAPE IN PHOTOREADER AND PRESS START TO INITIATE NEW SOLUTION.
```

MULTILAYER SOLUTION OF THE MODIFIED BERGGREN EQUATION

	DRY DENSITY (GAMMA)	WATER CONTENT (WC)	LAYER SIZE (D)	HEAT CAPACITY (C)	THERMAL COND. (THERM)	LATENT HEAT (FUSION)	FUSION PAR. (U)	LAMBDA (F)	LAYER RES. (RES)	CUM. RES. (SUMRES)	LAYER INDEX (AYERI)	CUM. INDEX (SI)
1	138.00	.00	.400	28.00	.8600	.	.00	.00	.47	.00	.	.
2	156.00	2.10	1.600	29.00	1.8500	470.	1.14	.45	.86	.47	139.	139.
3	151.00	2.80	3.000	29.00	2.0000	610.	.83	.50	1.50	1.33	632.	772.
4	130.00	6.50	1.000	28.00	1.6500	1220.	.67	.54	.61	2.83	552.	1323.
5	122.00	4.60	1.000	25.00	.6400	808.	.64	.55	1.56	3.44	477.	1800.
6	122.00	4.60	.496	25.00	.6400	808.	.65	.54	.78	3.44	217.	1541.
7	122.00	4.60	.540	25.00	.6400	808.	.65	.54	.84	3.44	238.	1562.

```
SURFACE INDEX =    1560. DEGREE DAYS F
N FACTOR =  2.00
LENGTH OF SEASON =   105. DAYS
MEAN ANNUAL TEMP =   12.0 F

DEPTH OF THAW PENETRATION FOR THIS SOLUTION  = 6.5 FEET.
```

Sample Solution No. 2

Typewriter output:

```
TYPE C, TAB, THERM, TAB, FUSION, CR.
28.  .86 Ø.
```

MULTILAYER SOLUTION OF THE MODIFIED BERGGREN EQUATION

	DRY DENSITY (GAMMA)	WATER CONTENT (WC)	LAYER SIZE (D)	HEAT CAPACITY (C)	THERMAL COND. (THERM)	LATENT HEAT (FUSION)	FUSION PAR. (U)	LAMBDA (F)	LAYER RES. (RES)	CUM. RES. (SUMRES)	LAYER INDEX (AYERI)	CUM. INDEX (SI)
1	138.00	.00	.400	28.00	.8600	.	.00	.00	.47	.00	.	.
2	156.00	2.10	1.600	28.98	1.8147	472.	1.13	.45	.88	.47	140.	140.
3	151.00	2.80	3.000	28.84	1.8502	609.	.83	.50	1.62	1.35	653.	793.
4	130.00	6.50	1.000	28.44	1.6460	1217.	.67	.54	.61	2.97	575.	1368.
5	122.00	4.60	2.000	24.95	.6296	808.	.61	.55	3.18	3.58	1133.	2502.
6	122.00	4.60	.339	24.95	.6296	808.	.66	.54	.54	3.58	151.	1519.
7	122.00	4.60	.431	24.95	.6296	808.	.66	.54	.69	3.58	194.	1563.

```
SURFACE INDEX =    1560. DEGREE DAYS F
N FACTOR =  2.00
LENGTH OF SEASON =  105. DAYS
MEAN ANNUAL TEMP =  12.0 F

DEPTH OF THAW PENETRATION FOR THIS SOLUTION  = 6.4 FEET.
```

Unclassified
Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY *(Corporate author)* Cold Regions Research & Engineering Laboratory U.S. Army Terrestrial Sciences Center, Hanover, New Hampshire	2a. REPORT SECURITY CLASSIFICATION Unclassified
	2b. GROUP

3. REPORT TITLE
DIGITAL SOLUTION OF MODIFIED BERGGREN EQUATION TO CALCULATE DEPTHS OF FREEZE OR THAW IN MULTILAYERED SYSTEMS

4. DESCRIPTIVE NOTES *(Type of report and inclusive dates)*
Special Report

5. AUTHOR(S) *(First name, middle initial, last name)*
George W. Aitken and Richard L. Berg

6. REPORT DATE	7a. TOTAL NO. OF PAGES	7b. NO. OF REFS
October 1968	22	4

8a. CONTRACT OR GRANT NO.	9a. ORIGINATOR'S REPORT NUMBER(S)
b. PROJECT NO.	Special Report 122
c.	9b. OTHER REPORT NO(S) *(Any other numbers that may be assigned this report)*
d.	

10. DISTRIBUTION STATEMENT
This document has been approved for public release and sale; its distribution is unlimited.

11. SUPPLEMENTARY NOTES	12. SPONSORING MILITARY ACTIVITY
	Office, Chief of Engineers Directorate of Military Construction Engineering Division Civil Engineering Branch

13. ABSTRACT
This report presents a method for a digital computer solution, using the FORTRAN language, of the modified Berggren equation for computing depths of frost and thaw penetration in non-homogeneous (multilayered) soil systems. A program source listing, sample solutions, and tables of thermal properties of soils and construction materials are presented.

DD FORM 1 NOV 65 1473 REPLACES DD FORM 1473, 1 JAN 64, WHICH IS OBSOLETE FOR ARMY USE.

Unclassified
Security Classification

Unclassified
Security Classification

14. KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT
Computer Heat flow Soil freezing						

Unclassified
Security Classification